

2.2 Numbers

Mathematica can work with arbitrarily long integer numbers.

In[7] := Factorial[100]

Out[7] = 933262154439441526816992388562667004907159682643816214685\
9296389521759999322991560894146397615651828625369792082722375\
8251185210916864000000000000000000000000

When working with a rational number, the greatest common divisors of its numerator and denominator are canceled.

In[8] := a = 1234567890/987654321

Out[8] = $\frac{137174210}{109739369}$

Calculations with rational numbers are exact.

In[9] := a^5

Out[9] = 48569355286282885522765185491603110100000/\
15915207065345784618237986236670245907849

How much is this numerically? Say, with 30 significant digits?

In[10] := N[a, 30]

Out[10] = 1.24999998860937500014238281250

Mathematica can work with real (floating-point) numbers having arbitrarily high precision.

In[11] := a = 1234567890987654321.1234567890987654321

Out[11] = 1.234567890987654321123456789098765432 $\times 10^{18}$

In[12] := a^5

Out[12] = 2.86797187177160567275921531725363508 $\times 10^{90}$

Here are π and e with 100 significant digits.

In[13] := N[Pi, 100]

Out[13] = 3.14159265358979323846264338327950288419716939937510582097\
4944592307816406286208998628034825342117068

In[14] := N[E, 100]

Out[14] = 2.71828182845904523536028747135266249775724709369995957496\
6967627724076630353547594571382178525166427

2.3 Polynomials and Rational Functions

Let's take a polynomial.

In[15] := a = (x + y + z)^6

Out[15] = $(x + y + z)^6$

Expand it.

In[16] := a = Expand[a]

$$\begin{aligned} \text{Out[16]} = & x^6 + 6x^5y + 15x^4y^2 + 20x^3y^3 + 15x^2y^4 + 6xy^5 + y^6 + 6x^5z + 30x^4yz + \\ & 60x^3y^2z + 60x^2y^3z + 30xy^4z + 6y^5z + 15x^4z^2 + 60x^3yz^2 + 90x^2y^2z^2 + \\ & 60xy^3z^2 + 15y^4z^2 + 20x^3z^3 + 60x^2yz^3 + 60xy^2z^3 + 20y^3z^3 + 15x^2z^4 + \\ & 30xyz^4 + 15y^2z^4 + 6xz^5 + 6yz^5 + z^6 \end{aligned}$$

The degree in x .

In[17] := Exponent[a, x]

$$\text{Out[17]} = 6$$

The coefficient of x^2 .

In[18] := Coefficient[a, x, 2]

$$\text{Out[18]} = 15y^4 + 60y^3z + 90y^2z^2 + 60yz^3 + 15z^4$$

Collect terms with the same power of x together.

In[19] := Collect[a, x]

$$\begin{aligned} \text{Out[19]} = & x^6 + y^6 + 6y^5z + 15y^4z^2 + 20y^3z^3 + 15y^2z^4 + 6yz^5 + z^6 + x^5(6y + 6z) + \\ & x^4(15y^2 + 30yz + 15z^2) + x^3(20y^3 + 60y^2z + 60yz^2 + 20z^3) + \\ & x^2(15y^4 + 60y^3z + 90y^2z^2 + 60yz^3 + 15z^4) + \\ & x(6y^5 + 30y^4z + 60y^3z^2 + 60y^2z^3 + 30yz^4 + 6z^5) \end{aligned}$$


Factorize it.

In[20] := a = Factor[a]

$$\text{Out[20]} = (x + y + z)^6$$

Suppose we want to factorize polynomials $x^n - 1$ with various n . The parameter n can be varied from 2 to 10 by dragging the marker with the mouse.

In[21] := Manipulate[Factor[x^n - 1], {n, 2, 10, 1, Appearance -> "Labeled"}]

Out[21] = 

There exists an algorithm which completely factorizes any polynomial with integer coefficients into factors which also have integer coefficients.

In[22] := Factor[x^4 - 1]

$$\text{Out[22]} = (-1 + x)(1 + x)(1 + x^2)$$

If we want to get factors whose coefficients come from an extension of the ring of integers, say, by the imaginary unit i , we should say so explicitly.

In[23] := Factor[x^4 - 1, Extension -> I]

$$\text{Out[23]} = (-1 + x)(-i + x)(i + x)(1 + x)$$

This polynomial factorizes into two factors with integer coefficients.

In[24] := a = x^4 - 4; Factor[a]

$$\text{Out[24]} = (-2 + x^2)(2 + x^2)$$

If coefficients from the extension of the ring of integers by $\sqrt{2}$ are allowed—into three factors.

In[25] := Factor[a, Extension -> Sqrt[2]]

$$\text{Out[25]} = -(\sqrt{2} - x)(\sqrt{2} + x)(2 + x^2)$$

And if the ring of coefficients is extended by both $\sqrt{2}$ and i —into four factors.

In[26] := Factor[a, Extension->{Sqrt[2], I}]

Out[26] = $-(\sqrt{2}-x)(\sqrt{2}-ix)(\sqrt{2}+ix)(\sqrt{2}+x)$

And this is a rational function.

In[27] := (x^3 - y^3)/(x^2 - y^2)

Out[27] = $\frac{x^3 - y^3}{x^2 - y^2}$

It is not canceled by the greatest common divisor of its numerator and denominator; this should be done explicitly.

In[28] := Cancel[%]

Out[28] = $\frac{x^2 + xy + y^2}{x + y}$

(% means the result of the previous calculation). A sum of rational functions.

In[29] := a = x/(x + y) + y/(x - y)

Out[29] = $\frac{y}{x - y} + \frac{x}{x + y}$

Let's put it over the common denominator.

In[30] := a = Together[a]

Out[30] = $\frac{x^2 + y^2}{(x - y)(x + y)}$

Partial fraction decomposition with respect to x .

In[31] := Apart[a, x]

Out[31] = $1 + \frac{y}{x - y} - \frac{y}{x + y}$

In[32] := Clear[a]

2.4 Elementary Functions

Mathematica knows some simple properties of elementary functions.

In[33] := Sin[-x]

Out[33] = $-\text{Sin}[x]$

In[34] := Cos[Pi/4]

Out[34] = $\frac{1}{\sqrt{2}}$

In[35] := Sin[5 * Pi/6]

Out[35] = $\frac{1}{2}$

In[36] := Log[1]

Out[36] = 0

In[37] := Log[E]

Out[37] = 1

In[38] := Exp[Log[x]]

Out[38] = x

In[39] := Log[Exp[x]]

Out[39] = $\text{Log}[e^x]$

And why not x ? Because this simplification is not always correct. Try to substitute $2\pi i$.

In[40] := **Sqrt**[0]

Out[40] = 0

In[41] := **Sqrt**[x]^2

Out[41] = x

In[42] := **Sqrt**[x^2]

Out[42] = $\sqrt{x^2}$

And why not x ? Try to substitute -1 .

In[43] := **a** = **Sqrt**[12 * x^2 * y]

Out[43] = $2\sqrt{3}\sqrt{x^2y}$

This result can be improved, if we know that $x > 0$.

In[44] := **Simplify**[a, x > 0]

Out[44] = $2\sqrt{3}x\sqrt{y}$

And this is the case $x < 0$.

In[45] := **Simplify**[a, x < 0]

Out[45] = $-2\sqrt{3}x\sqrt{y}$

Expansion of trigonometric functions of multiple angles, sums, and differences:

In[46] := **TrigExpand**[Cos[2 * x]]

Out[46] = $\text{Cos}[x]^2 - \text{Sin}[x]^2$

In[47] := **TrigExpand**[Sin[x - y]]

Out[47] = $\text{Cos}[y]\text{Sin}[x] - \text{Cos}[x]\text{Sin}[y]$

The inverse operation—transformation of products and powers of trigonometric functions into linear combinations of such functions—is used more often. Let's take a truncated Fourier series.

In[48] := **a** = **a1 * Cos[x] + a2 * Cos[2 * x] + b1 * Sin[x] + b2 * Sin[2 * x]**

Out[48] = $a_1 \text{Cos}[x] + a_2 \text{Cos}[2x] + b_1 \text{Sin}[x] + b_2 \text{Sin}[2x]$

Its square is again a truncated Fourier series.

In[49] := **TrigReduce**[a^2]

Out[49] =
$$\frac{1}{2} (a_1^2 + a_2^2 + b_1^2 + b_2^2 + 2a_1a_2\text{Cos}[x] + 2b_1b_2\text{Cos}[x] + a_1^2\text{Cos}[2x] - b_1^2\text{Cos}[2x] + 2a_1a_2\text{Cos}[3x] - 2b_1b_2\text{Cos}[3x] + a_2^2\text{Cos}[4x] - b_2^2\text{Cos}[4x] - 2a_2b_1\text{Sin}[x] + 2a_1b_2\text{Sin}[x] + 2a_1b_1\text{Sin}[2x] + 2a_2b_1\text{Sin}[3x] + 2a_1b_2\text{Sin}[3x] + 2a_2b_2\text{Sin}[4x])$$

2.5 Calculus

Let's take a function.

In[50] := **f** = **Log**[x^5 + x + 1] + 1/(x^5 + x + 1)

Out[50] = $\frac{1}{1+x+x^5} + \text{Log}[1+x+x^5]$

Calculate its derivative.

In[51] := g = D[f,x]

$$\text{Out[51]} = -\frac{1+5x^4}{(1+x+x^5)^2} + \frac{1+5x^4}{1+x+x^5}$$

Put over the common denominator.

In[52] := g = Together[g]

$$\text{Out[52]} = \frac{(1+5x^4)(x+x^5)}{(1+x+x^5)^2}$$

A stupid integration algorithm would try to solve the fifth degree equation in the denominator, in order to decompose the integrand into partial fractions. *Mathematica* is more clever than that.

In[53] := Integrate[g,x]

$$\text{Out[53]} = \frac{1}{1+x+x^5} + \text{Log}\left[1+x+x^5\right]$$

Let's expand our function in x at 0 up to x^{10} .

In[54] := Series[f,{x,0,10}]

$$\text{Out[54]} = 1 + \frac{x^2}{2} - \frac{2x^3}{3} + \frac{3x^4}{4} - \frac{4x^5}{5} + \frac{11x^6}{6} - \frac{20x^7}{7} + \frac{31x^8}{8} - \frac{44x^9}{9} + \frac{32x^{10}}{5} + O[x]^{11}$$

Mathematica can calculate many definite integrals even when the corresponding indefinite integral cannot be taken. Here is an integral from 0 to 1.

In[55] := Integrate[Log[x]^2/(x+1),{x,0,1}]

$$\text{Out[55]} = \frac{3 \text{Zeta}[3]}{2}$$

Mathematica knows how to sum many series.

In[56] := Sum[1/n^4,{n,1,Infinity}]

$$\text{Out[56]} = \frac{\pi^4}{90}$$

Let's clear all the garbage we have generated—a very good habit.

In[57] := Clear[f,g]

2.6 Lists

We have already encountered this construct several times:

In[58] := a = {x,y,z}

$$\text{Out[58]} = \{x,y,z\}$$

This is a list. And here are its elements.

In[59] := a[[1]]

$$\text{Out[59]} = x$$

In[60] := a[[2]]

$$\text{Out[60]} = y$$

In[61] := a[[3]]

$$\text{Out[61]} = z$$


In[62] := Clear[a]

2.7 Plots

A simple plot of a function.

In[63] := Plot[Sin[x]/x, {x, -10, 10}]


Out[63] =



A curve given parametrically— x and y are functions of t . This particular curve contains a parameter a , which can be adjusted by the mouse. If you click the small plus sign near the marker, a control panel will open. There you can start (and stop) animation.

In[64] := Manipulate[ParametricPlot[{Exp[a * t] * Cos[t], Exp[a * t] * Sin[t]}, {t, 0, 20}, PlotRange -> {{-10, 10}, {-10, 10}}, {a, 0.1}, 0, 0.2]]


Out[64] =



A three-dimensional plot of a function of two variables. It can be rotated by the mouse.

In[65] := Plot3D[x^2 + y^2, {x, -1, 1}, {y, -1, 1}]


Out[65] =



A three-dimensional curve given parametrically. The parameter a can be adjusted by the mouse.

**In[66] := Manipulate[ParametricPlot3D[{Cos[t], Sin[t], a * t}, {t, 0, 20},
PlotRange -> {{-1, 1}, {-1, 1}, {0, 2}}, {a, 0.1, 0, 0.2}]**


Out[66] =



A surface given parametrically.

In[67] := ParametricPlot3D[{Sin[t] * Cos[u], Sin[t] * Sin[u], Cos[t]}, {t, 0, Pi}, {u, 0, 2 * Pi}]

Out[67] =



2.8 Substitutions

Substitutions are a fundamental concept in *Mathematica*, its main working instrument. This substitution replaces $f[x]$ by x^2 .

In[68] := S = f[x] -> x^2

Out[68] = $f[x] \rightarrow x^2$

Let's apply it to the expression $f[x]$.

In[69] := f[x] /. S

Out[69] = x^2

We've got x^2 , as expected. And what if we apply it to $f[y]$?

In[70] := f[y] /. S

Out[70] = $f[y]$

It hasn't triggered. The following substitution replaces the function f with an arbitrary argument by the square of this argument.

In[71] := S = f[x.] -> x^2

Out[71] = f[x.] -> x^2

Let's check.

In[72] := {f[x], f[y], f[2]} /. S

Out[72] = {x^2, y^2, 4}

In[73] := Clear[S]

2.9 Equations

Here is an equation.

In[74] := Eq = a * x + b == 0

Out[74] = b + a x == 0

Let's solve it for x.

In[75] := S = Solve[Eq, x]

Out[75] = {{x -> -b/a}}

We've got a list of solutions, in this particular case having a single element. Each solution is a list of substitutions, which replaces our unknowns by the corresponding expressions. And how can we extract the value of x from this result? Let's take the first (and the only) element of the list S .

In[76] := S1 = First[S]

Out[76] = {x -> -b/a}

And now we apply this list of substitutions (in this particular case, it's single element) to the unknown x .

In[77] := x /. S1

Out[77] = -b/a

Here is a more advanced example—a quadratic equation. It has two solutions.

In[78] := S = Solve[a * x^2 + b * x + c == 0, x]

Out[78] = {{x -> (-b - Sqrt[b^2 - 4ac])/2a}, {x -> (-b + Sqrt[b^2 - 4ac])/2a}}

How can we extract the value of x in the second solution? Let's apply the second element of the solutions list S (which is a single-element list of substitutions) to the unknown x .

In[79] := x /. S[[2]]

Out[79] = (-b + Sqrt[b^2 - 4ac])/2a

And here is a system of 2 linear equations.

In[80] := Eq = {a * x + b * y == e, c * x + d * y == f}

Out[80] = {a x + b y == e, c x + d y == f}

It has a single solution.

In[81] := S = Solve[Eq, {x,y}]

$$\text{Out[81]} = \left\{ \left\{ x \rightarrow -\frac{de - bf}{bc - ad}, y \rightarrow -\frac{-ce + af}{bc - ad} \right\} \right\}$$

This (first and the only) solution is a list of two substitutions.

In[82] := S1 = S[[1]]

$$\text{Out[82]} = \left\{ x \rightarrow -\frac{de - bf}{bc - ad}, y \rightarrow -\frac{-ce + af}{bc - ad} \right\}$$

How to find the values of x and y in this solution? Apply this list of substitutions to the unknowns x and y .

In[83] := {x/.S1,y/.S1}

$$\text{Out[83]} = \left\{ -\frac{de - bf}{bc - ad}, -\frac{-ce + af}{bc - ad} \right\}$$

In[84] := Clear[Eq,S,S1]



<http://www.springer.com/978-3-319-00893-6>

Introduction to Mathematica® for Physicists

Grozin, A.

2014, X, 219 p., Hardcover

ISBN: 978-3-319-00893-6